

# MATHEMATICAL MODEL OF THE HAY AGGREGATOR FELLA TS 1602

MATEMATICKÝ MODEL DYNAMIKY SHRNOVAČE PÍCE  
FELLA TS 1602

**Šmíd V., Bartoň S.**

Department of Engineering and Automobile Transport, Faculty of Agronomy, Mendel University of Agriculture and Forestry in Brno, Zemědělská 1, 613 00, Brno, Czech Republic

E-mail: vlasmi@centrum.cz, barton@mendelu.cz

---

## ABSTRACT

The work presents construction of mathematical model of the hay aggregator. Analytical model is based on the programme Maple 11 and enables evaluation of the influence of the dimensions of constructional parts on the kinematics and dynamics of the whole machine. This approach enables creation of precise animations showing movement of the key constructional parts, computation and drawing of precise graphical dependencies showing dynamic stress or consumption of energy. Study of the model enables detection of the constructional imperfections and demonstrates possibility of improvements.

**Key words:** Maple 11, mathematical modelling, analytical description, animation, construction, quality evaluation

**Acknowledgments:** Authors would like to express their grateful thanks to the company TOKO-AGRO a.s. Rudice u Luhačovic for kind lending of the constructional parts of the hay aggregator.

## ÚVOD

Pícniny jsou velmi důležitou plodinou z hlediska zajištění krmivové základny hospodářských zvířat a základním zdrojem objemných krmiv. Tvoří je travní porosty z luk a pastvin, dále víceleté pícniny – jeteloviny (vojtěška, jetel), pícní trávy a jednoleté pícniny (zvláště kukuřice a směsky) pěstované na orné půdě.

Sklizeň pícnin probíhá téměř pro celé vegetační období se špičkami v době prvních sečí – senoseče a sklizně silážních plodin. Hlavním problémem při sklizni je zmenšit riziko počasí a tím snížit sklizňové a konzervační ztráty. Při špatném počasí a nevhodném způsobu sklizně, odrolem, nesebráním, nevhodnou konzervací mohou činit ztráty sušiny na hmotě 15 až 35 %, ztráty živin až 50 % a vitamínů až 100 %. Vhodným sklizňovým pracovním postupem a konzervací lze snížit riziko počasí a zabránit znehodnocení píce v průběhu uskladnění.

Základními články technologické linky pro sklizeň píce jsou žací stroj s případnou úpravou pokosu – obraceč dle potřeby – shrnovač – sběrací vůz, lis, nebo řezačka.

I když se to zdá jako samozřejmost, tak v rámci sklizňových prací je shrnování zavadlé hmoty na řádek důležitým krokem. Pro maximální využití optimálního obsahu sušiny musí práce probíhat s velkou výkonností umožňující souběžný provoz shrnovače a řezačky (nebo jiného sklizňového stroje). Rozhodující faktory pro práci shrnovače jsou šířka záběru a s tím spojené množství hmoty spolu s charakterem ukládaného řádku. Shrnování píce do objemných řádků umožňuje i při nižší pojezdové rychlosti optimálně využít instalovaného výkonu moderních lisů, senážních návěsů a řezaček. Menší počet řádků s větším množstvím píce snižuje počet přejezdů těchto strojů na pozemku a podíl ztrátových časů pro otáčení.

Řádek s rovnoměrným ukládáním hmoty snižuje nárazová zatížení sklizňových strojů a jejich výkon je vyšší. Ideální řádek není okrouhlý, nýbrž téměř pravoúhlý a na okrajích je větší množství hmoty než ve středu. V průřezu má tvar písmene M.

Je zřejmé, že šířka odkládaného řádku by měla odpovídat záběru sběrače následného stroje. Méně známé je ale to, že šířka řádku by měla být v tomto intervalu co největší, aby se dosáhlo rovnoměrného zatížení plnicího ústrojí.

U shrnovačů by mělo být účinné kopírování terénu podvozky rotorů a precizní nastavení výšky by mělo sloužit především k tomu, aby nedocházelo k poškozování travního drnu a zvýšenému znečišťování píce. Kvalitního kopírování terénu je dosaženo kardanovým zavěšením rotorů. Každý rotor se přizpůsobí terénu nezávisle jak v podélném, tak i příčném směru.

Dvourotorové shrnovače se středovým ukládáním hmoty jsou známé tím, že vytvářejí velmi rovnoměrné řádky s nízkým znečištěním díky krátké vzdálenosti příčného posunu hmoty. Se dvěma rotory se je možné dosáhnout záběru jen něco přes 8 metrů, takže podmínkou využití výkonu řezačky je dostatečný výnos hmoty. Naproti tomu dvourotorové shrnovače s bočním ukládáním řádků dovolují dvě nebo více jízd složit do jednoho řádku. V tomto případě lze v jednotlivých sečích formovat řádky adekvátní sklizňové technice bez přímé závislosti na výnosu hmoty.

Úspěšnou kombinaci vysoké výkonnosti s ideálním formováním řádku představují čtyřrotorové shrnovače se středovým ukládáním. Plynule nastavitelná šířka záběru až 13,5 m vytváří předpoklady pro vysoký výkon 8 až 10 ha.h<sup>-1</sup> při nízkém počtu přejezdů s minimem ztrátových časů při obracení. Časový náskok, s nímž musí vyrazit do práce, je minimální a tím lze využít poslední cenné hodiny pro zavádání shrnované hmoty. Výjimkou nejsou ani situace, kdy tento shrnovač pracuje těsně před lisem nebo řezačkou. Čas pro zavádání je maximálně využitý a dosahuje se velmi rovnoměrného obsahu sušiny. Ohlasy z praxe potvrzují zvýšení výkonu řezačky až o 30 % díky rovnoměrnosti formovaných řádků v porovnání s malými shrnovači. Ze stejného důvodu tyto shrnovače dobře zpracují hmotu i pro lisy či senážní návěsy. V některých případech se možnost náhrady dvojice menších shrnovačů se záběrem až 7 m ukázala jako výhodná z důvodu omezení počtu přejezdů na citlivých pozemcích.

Tyto shrnovače se prosadily nejprve na velkých pozemcích, ale nyní se objevují i v kopcovitých oblastech s členitými pozemky. A to proto, že s nástupem moderních senážních návěsů se stávající dvurotorové shrnovače staly nejslabším místem sklizně. Bylo často nutné zároveň povolat ještě druhý stroj a začínat se shrnováním s více než půldenním předstihem před startem sběru hmoty. Čtyřrotorový shrnovač dovoluje téměř souběžný začátek práce celé linky. Při blížící se změně počasí je možné píci rychle shrnout a nařezat, což je důležité v regionech s častými dešti a pěti sečemi do roka.

Při velkém výnosu hmoty na pozemku jsou ovšem formované řádky příliš vysoké, takže činí problémy sběrači senážních návěsů. Za této situace je vhodnější využití dvojrotorového provedení. To samé platí i pro sklizeň luk s přerostlou trávou ke krmení koní.

Proto se v podnicích zatím nejvíce prosadily univerzálnější vysoko záběrové dvojrotorové shrnovače s bočním nahrnováním, které umožňují při malém výnosu hmoty ve dvou protiběžných jízdách nahrnout 2 řádky na sebe a tím dostatečně vytížit odvozový prostředek. Při vysokých výnosech zase nevytvářejí příliš vysoký řádek

Proto se ve své práci zabývám matematickým modelem dynamiky jednoho ramene shrnovače píce FELLA TS 1602 Hydro, typického zástupce této kategorie.



Obr. 1 Shrnovač Fella TS 1602 Hydro



Obr. 2 Detail shrnovacího rotoru

## MATERIÁL A METODIKA

### Fella TS 1602

Jedná se o výrobek firmy FELLA – Werke GmbH & Co. KG, Feucht, Deutschland. Je to tažený dvourotorový stranový shrnovač s nastavitelným záběrem 7,0 – 8,4 m umožňující shrnování do jednoho nebo dvou řádků. Rotory mají průměr 4,2 m, první rotor je vybaven 10 shrnovacími rameny, druhý 12 rameny, shrnovací výška je nezávisle nastavitelná mechanicky klikou. Každý rotor má vlastní čtyřkolový podvozek. Natáčení shrnovacích ramen do a z pracovní polohy je zajištěno pohybem vodících roln v křivkové dráze – *vedení vačky*, ze

zušlechťené a otěruvzdorné litiny. Stroj je v zadní části vybaven tandemovou nápravou řízenou od výkyvné ojové hlavy prostřednictvím otočného hřídele uloženého v hlavním nosníku. Potřebný výkon tažného prostředku 30 kW, otáčky vývodového hřídele  $540 \text{ min}^{-1}$  celková hmotnost 1950 kg.

### Souřadný systém pro tvorbu modelu

Tvorba matematického modelu bude nejjednodušší v systému kartézských souřadnic  $E3$ . Počátek souřadného systému je umístěn do středu vodící dráhy ramene vačky tak, že  $Z$  je totožná s osou rotace zařízení. Osa  $Y$  je orientovaná ve směru pojezdu sběrače a osa  $X$  bude na osy  $Z$  a  $Y$  kolmá, kladný směr je vpravo ve směru pojezdu.

Pro výpočet byl použit matematický program MAPLE 11. Z důvodů zvýšení přehlednosti budou uváděny pouze nejdůležitější výstupy výpočtů – odpovědi programu MAPLE na zadávané příkazy.

### Základy matematického modelu

Zavedeme funkční substitute za  $z$  souřadnici vedení vačky, jejich použití zvyšuje přehlednost výpočtu, protože  $z$  souřadnice bude složitou, po částech definovanou funkcí.

```
> restart;
> with(plots):>
DSu:=[diff(z(t),t)=Zt,diff(z(t),t,t)=Ztt,z(t)=Z];
```

$$DSu := \left[ \frac{d}{dt} z(t) = Zt, \frac{d^2}{dt^2} z(t) = Ztt, z(t) = Z \right]$$

Nyní zavedeme konstrukční substitute:

- $r$  = délka vačky – 0,1 m
- $R$  = základní poloměr vodící křivkové dráhy – 0,5 m
- $T$  = perioda rotace –  $1 \text{ s}^{-1}$
- $L$  = délka shrnovacího ramene – 1,5 m
- $l$  = délka shrnovacího prstu – 0,4 m
- $V$  = pojezdová rychlost soupravy –  $10 \text{ km} \cdot \text{h}^{-1}$
- $Mt$  = hmotnost shrnovacího ramene – 5 kg
- $Mp$  = hmotnost shrnovacího prstu – 1 kg
- $dp$  = vzájemná rozteč shrnovacích prstů – 0,1 m
- $n$  = počet shrnovacích prstů - 4

```
> KSu:=[r=1/10,R=1/2,T=1,L=3/2,l=4/10,V=120/36,Mt=5,Mp=1,dp=1/10,n=4];
```

$$KSu := \left[ r = \frac{1}{10}, R = \frac{1}{2}, T = 1, L = \frac{3}{2}, l = \frac{4}{10}, V = \frac{10}{3}, Mt = 5, Mp = 1, dp = \frac{1}{10}, n = 4 \right]$$

Pracovní rameno se otáčí o  $\pm 45^\circ$  z neutrální polohy, proto  $\alpha = 45^\circ$ ;  $\omega$  je úhlová rychlost otáčení nosné tyče okolo vertikální osy.

```
> alpha:=Pi/4; omega:=2*Pi/T;
```

$$\alpha := \frac{\pi}{4} \qquad \omega := \frac{2\pi}{T}$$

Vertikální souřadnice vedení vačky –  $z$  je po částech definovaná funkce v závislosti na úhlu pootočení nosné tyče. Tento úhel je funkcí času, proto je  $z$  přímo definovaná jako periodická funkce času s periodou  $T$ .

```
> Z:=simplify(r*sin(alpha)*
piecewise(t-floor(t/T)*T<T/4,-1,
```


```
t-floor(t/T)*T<3*T/8,sin((t+7*T/16)*4*omega),
t-floor(t/T)*T<5/8*T,1,
t-floor(t/T)*T<3*T/4,sin((t-1*T/16)*4*omega),
t-floor(t/T)*T<T,-1));
```

```
> Zt:=simplify(diff(Z,t)): Ztt:=simplify(diff(Z,t,t)):
```

$$Z = \frac{1}{2} r \sqrt{2} \begin{cases} -1 & t - \text{floor}\left(\frac{t}{T}\right)T < \frac{T}{4} \\ \sin\left(\frac{(16t+7T)\pi}{2T}\right) & t - \text{floor}\left(\frac{t}{T}\right)T < \frac{3T}{8} \\ 1 & t - \text{floor}\left(\frac{t}{T}\right)T < \frac{5T}{8} \\ \sin\left(\frac{(16t-T)\pi}{2T}\right) & t - \text{floor}\left(\frac{t}{T}\right)T < \frac{3T}{4} \\ -1 & t - \text{floor}\left(\frac{t}{T}\right)T < T \end{cases}$$

Pro názornost je možné vykreslit průběh souřadnice  $z$  i jejích derivací podle času.

```
> plot(subs(DSu,KSu,[z(t),diff(z(t),t)/8/Pi,
diff(z(t),t,t)/64/Pi^2]),t=0..2,color=[black,red,blue],
thickness=2,numpoints=1000,
title="Prubeh suradnice z a jejich derivaci podle casu");
```



Obr. 3 Průběh souřadnice  $z$  a jejích derivací podle času

Vzhledem k tomu, že opěrné rameno tyče se musí pohybovat po kružnici o poloměru  $R$ , musí být  $\rho$  – vzdálenost vedení vačky od osy rotace proměnlivá, Tato proměnná může být využita pro výpočet  $x$  a  $y$  souřadnice vedení vačky.

```
> rho:=simplify(R+r-sqrt(r^2-z(t)^2),symbolic):
```

```
> x:=rho*cos(omega*t); y:=rho*sin(omega*t);
```

$$x := (R + r - \sqrt{r^2 - z(t)^2}) \cos\left(\frac{2\pi t}{T}\right) \quad y := (R + r - \sqrt{r^2 - z(t)^2}) \sin\left(\frac{2\pi t}{T}\right)$$

Nyní je nutné vytvořit proměnnou, která bude popisovat tvar vodicí dráhy vačky a vedení ramene vačky.


```
> Vacka:=[x,y,z(t)]:
```

```
> Vedeni:=[(R+r)*cos(omega*t),(R+r)*sin(omega*t),0];
```

$$\text{Vedeni} := \left[ (R+r) \cos\left(\frac{2\pi t}{T}\right), (R+r) \sin\left(\frac{2\pi t}{T}\right), 0 \right]$$

Výsledky výpočtů je pro dané konstrukční parametry možné znázornit graficky – jako prostorové křivky

```
> P1:=plots[spacecurve](subs(DSu,KSu,Vacka),
  t=0..1,color=black,thickness=3,numpoints=500):
> P2:=plots[spacecurve](subs(DSu,KSu,Vedeni),t=0..1,
  color=red,thickness=3,numpoints=500):
> plots[display]({P1,P2}, title="Trajektorie vacky a ramene",
  axes=boxed,orientation=[65,40],labels=["X","Y","Z"]);
```



Obr. 4 Trajektorie vacky a ramene: černě - prostorová křivka po níž běhá vačka, červeně - křivka po níž běží rameno vačky

Úhel stoupání vedení vačky, který je důležitým parametrem určujícím dynamické namáhání, je možné vypočítat z úhlu, který svírá tečný vektor k vedení vačky  $T\mathbf{v}$  s osou  $Z$ . Pro výpočet bude nutné dále znát absolutní velikost vektoru –  $AT\mathbf{v}$  a definovat jednotkový vektor ve směru osy –  $V\mathbf{v}$ . Úhel je přepočten na úhlové stupně a výsledek znázorněn graficky.


```
> Tv:=factor(diff(Vacka,t));
```

$$T\mathbf{v} := \left[ \left( z(t) \left( \frac{d}{dt} z(t) \right) \cos\left(\frac{2\pi t}{T}\right) T - 2 \sin\left(\frac{2\pi t}{T}\right) \pi \sqrt{(r-z(t))(r+z(t))} R \right. \right. \\ \left. \left. - 2 \sin\left(\frac{2\pi t}{T}\right) \pi \sqrt{(r-z(t))(r+z(t))} r + 2 \sin\left(\frac{2\pi t}{T}\right) \pi r^2 - 2 \sin\left(\frac{2\pi t}{T}\right) \pi z(t)^2 \right) \right. \\ \left. / \left( \sqrt{(r-z(t))(r+z(t))} T \right), - \left( -z(t) \left( \frac{d}{dt} z(t) \right) \sin\left(\frac{2\pi t}{T}\right) T \right. \right. \\ \left. \left. - 2 \cos\left(\frac{2\pi t}{T}\right) \pi \sqrt{(r-z(t))(r+z(t))} R - 2 \cos\left(\frac{2\pi t}{T}\right) \pi \sqrt{(r-z(t))(r+z(t))} r \right. \right. \\ \left. \left. + 2 \cos\left(\frac{2\pi t}{T}\right) \pi r^2 - 2 \cos\left(\frac{2\pi t}{T}\right) \pi z(t)^2 \right) / \left( \sqrt{(r-z(t))(r+z(t))} T \right), \frac{d}{dt} z(t) \right]$$

```

> ATv:=sqrt(sum(Tv[i]^2,i=1..3)): Vv:=[0,0,1];
> Stoupani:=convert(Pi/2-arccos(Tv[3]/ATv),degrees)/degrees:
> plot(subs(DSu,KSu,Stoupani),t=0..1,title="Uhel stoupani ve
stupnich");

```



Obr. 5 Průběh úhlu stoupaní vedení vačky


Maximální stoupaní je pro  $5/16$  periody, pro zajímavost je možné znázornit, jak závisí tento úhel na konstrukčních parametrech  $r$  a  $R$ . Toto zobrazení vyžaduje provedení dvou pomocných substitucí.

```

> Su2:=[z(t)='Z',alpha=Pi/4,omega=2*Pi,T=1]: Su3:=t=5/16:
> P31:=plot3d(subs(Su3,eval(subs(Su2[],Stoupani))),R=0.1..0.8,
r=0.01..0.2,color=black,style=hidden):
> P32:=contourplot3d(subs(Su3,eval(subs(Su2[],Stoupani))),
R=0.1..0.8,r=0.01..0.2,contours=5*[$1..15],color=black,
thickness=3,grid=[30,30]):
> display({P31,P32},axes=boxed,orientation=[-45,45],
title="Uhel maximalniho stoupani v zavislosti na r a R \n
Isolinie po 5 deg");

```





Obr. 6 Úhel maximálního stoupání v závislosti na délce vačky a poloměru vodící křivkové dráhy. Silnou čarou znázorněny isolinie stoupání s krokem 5 stupňů

## Model kinematiky

Nejprve je nutné vytvořit proměnné *Pojezd* – vektor popisující směr pojezdu, dále proměnnou *Tyc* – udávající polohu volného konce nosného ramene a *A* – úhel pootočení nosné tyče okolo její podélné osy, tedy úhel naklonění pracovních konců prstů.

```
> Pojezd := [0, V*t, 0]; Alpha := arccos(z(t)/r) - Pi/4;
> Tyc := Vedeni + [L*sin(omega*t), -L*cos(omega*t), 0];
```

$$Pojezd := [0, V t, 0] \quad A := \arccos\left(\frac{z(t)}{r}\right) - \frac{\pi}{4}$$

$$Tyc := \left[ L \sin\left(\frac{2 \pi t}{T}\right) + (R + r) \cos\left(\frac{2 \pi t}{T}\right), -L \cos\left(\frac{2 \pi t}{T}\right) + (R + r) \sin\left(\frac{2 \pi t}{T}\right), 0 \right]$$

Nejprve se provede znázornění pracovního cyklu jedné nosné tyče se čtyřmi pracovními prsty. Tyč bude vykreslena v 18ti polohách. Proto je nutné vytvořit proměnnou *T*, která rozdělí periodu oběhu na odpovídající časové okamžiky.

```
> Nf := 18;
> Tau := subs(KSu, T) / Nf * [$0..Nf]:
```

Nyní je nutné dopočítat polohy horních i dolních konců pracovních prstů, provést dosazení za obecnou funkci *z* a jejich derivací podle času a na závěr dosadit konkrétní časové hodnoty.

```
> for j from 1 to n do;
  Tyc[j] := Vedeni + [(L-dp*(j-1))*sin(omega*t),
    -(L-dp*(j-1))*cos(omega*t), 0];
  Prst[j] := Tyc[j] - [l*cos(omega*t)*cos(Alpha),
    l*sin(omega*t)*cos(Alpha), l*sin(Alpha)];
  atyc[j] := subs(DSu, KSu, [Vedeni, Tyc[j]]);
  aprst[j] := subs(DSu, KSu, [Tyc[j], Prst[j]]);
  rtyc[j] := evalf([seq(atyc[j], t=Tau)], 5);
  rprst[j] := evalf([seq(aprst[j], t=Tau)], 5);
end do;
```

Pro názornost je možné vypsát souřadnice konců 1. prstu – horní konec –  $Tyc_1$ , dolní konec –  $prst_1$

```
> 'Tyc[1]'=Tyc[1];'Prst[1]'=Prst[1];
```

$$Tyc_1 = \left[ L \sin\left(\frac{2\pi t}{T}\right) + (R+r) \cos\left(\frac{2\pi t}{T}\right), -L \cos\left(\frac{2\pi t}{T}\right) + (R+r) \sin\left(\frac{2\pi t}{T}\right), 0 \right]$$

$$Prst_1 = \left[ -l \cos\left(\frac{2\pi t}{T}\right) \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) + L \sin\left(\frac{2\pi t}{T}\right) + (R+r) \cos\left(\frac{2\pi t}{T}\right), \right. \\ \left. -l \sin\left(\frac{2\pi t}{T}\right) \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) - L \cos\left(\frac{2\pi t}{T}\right) + (R+r) \sin\left(\frac{2\pi t}{T}\right), \right. \\ \left. l \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \right]$$

Stejně operace je nutné provést i s vedením vačky a trajektorií ramene vačky a spojit vypočtené polohy prstů v jeden celek,

```
> avacka:=subs(DSu,KSu,[Vacka,Vedeni]):
```

```
> rvacka:=[seq(avacka,t=Tau)]:
```

```
> P3v:=spacecurve({rvacka[]},color=blue,thickness=3):
```


```
> P3t:=spacecurve({rtyc[1][]},color=green,thickness=3):
```

```
> P3p:=spacecurve({seq(rprst[j][],j=1..n)},color=brown,thickness=3):
```

```
> P3:=display({P3v,P3t,P3p}):
```

Na závěr je již možné vykreslit celkové oběhové schéma.

```
> display({P1,P2,P3},scaling=constrained, title="Schéma pracovního cyklu", axes=boxed,labels=["X","Y","Z"]);
```



Obr. 7 Schéma pracovního cyklu

## Animace

Tvorba animace se v podstatě neliší od vytvoření oběhového schématu. Pouze je nutné vzít v úvahu pohyb celé soustavy vlivem pohybu za traktorem, který popisuje proměnná *Pojezd*. Části, které se pohybují pouze vlivem pojezdu, budou vykresleny jako statické prostorové křivky ve 100 bodech a poté se vlivem pojezdu budou posouvat vpřed bez dalších změn. Animace bude znázorňovat dva pracovní cykly složené z 240 dílčích obrázků.

```

> asnek:=subs(DSu,KSu,Vacka):
  avedeni:=subs(DSu,KSu,Vedeni):
  avacka:=subs(DSu,KSu,[Vacka+Pojezd,Vedeni+Pojezd]):
> Np:=100:
> Tau:=subs(KSu,T)/Np*[$0..Np]:
> csnek:=[evalf(seq(asnek,t=Tau))]:
> cvedeni:=[evalf(seq(avedeni,t=Tau))]:
> Tau:=subs(KSu,T)/Nf*[$0..2*Nf]:
> rpojezd:=[seq(subs(KSu,Pojezd),t=Tau)]:
> rsnek:=map(u->map(v->u+v,csnek),rpojezd):
> rvedeni:=map(u->map(v->u+v,cvedeni),rpojezd):
> rvacka:=[seq(avacka,t=Tau)]:
> for j from 1 to n do;
  atyc[j]:=subs(DSu,KSu,[Vedeni+Pojezd,Tyc[j]+Pojezd]):
  aprst[j]:=subs(DSu,KSu,[Tyc[j]+Pojezd,Prst[j]+Pojezd]):
  rtyc[j]:=evalf([seq(atyc[j],t=Tau)],5):
  rprst[j]:=evalf([seq(aprst[j],t=Tau)],5):
  tprst[j]:=subs(DSu,KSu,Prst[j]+Pojezd):
  Trprst[j]:=[evalf(seq(tprst[j],t=Tau),5)]:
  Trprst[j]:=[Trprst[j][1],Trprst[j][]]:
end do:


```

Nyní je možné vytvořit animaci jednotlivých konstrukčních částí, které po vykreslení pohromadě vytvoří finální animaci celého zařízení.

```

> A_vacka:=display([seq(spacecurve(G,color=blue,thickness=3),
  G=rvacka)],insequence=true):
> A_tyc:=display([seq(spacecurve(G,color=green,thickness=3),
  G=rtyc[1])],insequence=true):
> A_prst:=display([seq(spacecurve({seq(rprst[i][j],i=1..n)},
  color=brown,thickness=3),j=1..2*Nf+1)],insequence=true):
> A_snek:=display([seq(spacecurve(G,color=black,thickness=3),
  G=rsnek)],insequence=true):
> A_vedeni:=display([seq(spacecurve(G,color=red,thickness=3),
  G=rvedeni)],insequence=true):
> A_Trprst:=display([seq(spacecurve({seq(Trprst[i][1..j+1],
  i=1..n)},color=red),j=1..2*Nf+1)],insequence=true
> display({A_vacka,A_tyc,A_prst,A_snek,A_vedeni,A_Trprst},
  axes=boxed,labels=["X","Y","Z"],title="Animace cinnosti");


```



Obr. 8 Animace činnosti shrnovače – odkaz na externí přehrávač animací

Pro názornost je možné vykreslit prostorovou křivku znázorňující trajektorii pracovních konců všech 4 prstů v časovém intervalu 5ti period.

```
> spacecurve({seq(tprst[j],j=1..n)},t=0..5,color=black,
  thickness=3,numpoints=1000,axes=boxed,labels=["X","Y","Z"]);
```




Obr.9 Trajektorie pohybu konců shrnovacích prstů

## Dynamika


Konkrétní výpočty dynamiky je nutné provádět pro zadané konstrukční parametry a pro určité časové okamžiky. Proto bude jedna pracovní perioda rozdělena na 500 časových kroků. Nejprve se vypočtou a graficky znázorní v odpovídajících vektorových prostorech vektory rychlostí a zrychlení pracovních konců prstů.

```
> Np:=500:
Tau:=1/Np*[$0..Np]:
> for j from 1 to n do;
  V_prst[j]:=diff(Prst[j],t)+[0,V,0];
  V_Ps[j]:=subs(DSu[],KSu,V_prst[j]):
  V_P[j]:=[seq(evalf(V_Ps[j]),t=Tau)]:
  A_prst[j]:=diff(Prst[j],t,t);
  A_Ps[j]:=subs(DSu,KSu,A_prst[j]):
  A_P[j]:=[seq(evalf(A_Ps[j]),t=Tau)]:
end do:
> spacecurve({seq(V_P[j],j=1..n)},color=black,thickness=3,
  axes=boxed,labels=[Vx,Vy,Vz],title="Rychlost koncu prstu");
```



Obr. 10 Velikost rychlosti konců prstů v jednotlivých osách

```
> spacecurve({seq(A_P[j], j=1..n)}, color=black, thickness=3,
  axes=boxed, labels=[Ax, Ay, Az], title="Zrychlení koncu prstu");
```



Obr. 11 Velikost zrychlení konců prstů v jednotlivých osách


Pro názornost je zde uveden výstup programu MAPLE udávající vektor rychlosti a zrychlení 1. pracovního prstu. Z důvodů úspory místa je velikost výstupu redukována.

> 'V\_prst[1]'=V\_prst[1]; 'A\_prst[1]'=A\_prst[1];

$$\begin{aligned}
 V_{prst_1} = & \left[ \frac{2 l \sin\left(\frac{2 \pi t}{T}\right) \pi \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right)}{T} + \frac{l \cos\left(\frac{2 \pi t}{T}\right) \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)}{r \sqrt{1 - \frac{z(t)^2}{r^2}}} + \frac{2 L \cos\left(\frac{2 \pi t}{T}\right) \pi}{T} \right. \\
 & - \frac{2 (R+r) \sin\left(\frac{2 \pi t}{T}\right) \pi}{T}, V - \frac{2 l \cos\left(\frac{2 \pi t}{T}\right) \pi \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right)}{T} + \frac{l \sin\left(\frac{2 \pi t}{T}\right) \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)}{r \sqrt{1 - \frac{z(t)^2}{r^2}}} + \frac{2 L \sin\left(\frac{2 \pi t}{T}\right) \pi}{T} \\
 & \left. + \frac{2 (R+r) \cos\left(\frac{2 \pi t}{T}\right) \pi}{T}, \frac{l \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)}{r \sqrt{1 - \frac{z(t)^2}{r^2}}} \right] \\
 A_{prst_1} = & \left[ \frac{4 l \cos\left(\frac{2 \pi t}{T}\right) \pi^2 \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right)}{T^2} - \frac{4 l \sin\left(\frac{2 \pi t}{T}\right) \pi \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)}{T r \sqrt{1 - \frac{z(t)^2}{r^2}}} \right. \\
 & + \frac{l \cos\left(\frac{2 \pi t}{T}\right) \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)^2}{r^2 \left(1 - \frac{z(t)^2}{r^2}\right)} + \frac{l \cos\left(\frac{2 \pi t}{T}\right) \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d^2}{dt^2} z(t)\right)}{r \sqrt{1 - \frac{z(t)^2}{r^2}}} \\
 & + \frac{l \cos\left(\frac{2 \pi t}{T}\right) \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)^2 z(t)}{r^3 \left(1 - \frac{z(t)^2}{r^2}\right)^{(3/2)}} - \frac{4 L \sin\left(\frac{2 \pi t}{T}\right) \pi^2}{T^2} - \frac{4 (R+r) \cos\left(\frac{2 \pi t}{T}\right) \pi^2}{T^2}, \frac{4 l \sin\left(\frac{2 \pi t}{T}\right) \pi^2 \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right)}{T^2} \\
 & + \frac{4 l \cos\left(\frac{2 \pi t}{T}\right) \pi \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)}{T r \sqrt{1 - \frac{z(t)^2}{r^2}}} + \frac{l \sin\left(\frac{2 \pi t}{T}\right) \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)^2}{r^2 \left(1 - \frac{z(t)^2}{r^2}\right)} \\
 & + \frac{l \sin\left(\frac{2 \pi t}{T}\right) \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d^2}{dt^2} z(t)\right)}{r \sqrt{1 - \frac{z(t)^2}{r^2}}} + \frac{l \sin\left(\frac{2 \pi t}{T}\right) \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)^2 z(t)}{r^3 \left(1 - \frac{z(t)^2}{r^2}\right)^{(3/2)}} + \frac{4 L \cos\left(\frac{2 \pi t}{T}\right) \pi^2}{T^2} \\
 & - \frac{4 (R+r) \sin\left(\frac{2 \pi t}{T}\right) \pi^2}{T^2}, \frac{l \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)}{r^2 \left(1 - \frac{z(t)^2}{r^2}\right)} + \frac{l \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d^2}{dt^2} z(t)\right)}{r \sqrt{1 - \frac{z(t)^2}{r^2}}} + \frac{l \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)^2 z(t)}{r^3 \left(1 - \frac{z(t)^2}{r^2}\right)^{(3/2)}} \left. \right]
 \end{aligned}$$


Dále je možné vykreslit závislost absolutní velikosti vektoru rychlosti a zrychlení pracovních prstů na čase.

```
> plot({seq(zip((u,v)->[u,sqrt(v[1]^2+v[2]^2+v[3]^2)],
  Tau,V_P[j]),j=1..n)},thickness=3,color=red,
  title="Absolutni velikost rychlosti koncu prstu",
  labels=["t","v"]);
```



Obr. 12 Absolutní velikost rychlosti konců prstů v průběhu jedné otáčky

```
> plot({seq(zip((u,v)->[u,sqrt(v[1]^2+v[2]^2+v[3]^2)],
  Tau,A_P[j]),j=1..n)},thickness=3,color=red,
  title="Absolutni velikost zrychleni koncu prstu",
  labels=["t","A"]);
```




Obr. 13 Absolutní velikost zrychlení konců prstů v průběhu jedné otáčky

Výkon je možné vypočítat jako skalární součin vektorů síly a zrychlení. V tomto případě by síla závisela na hmotnosti nabraného materiálu a hmotnostech pohybujících se částí, tzn. že výkon – příkon bude přímo úměrný velikosti skalárního součinu vektorů rychlosti a zrychlení.

```

> Vykon:=[seq(zip((u,v)->u[1]*v[1]+u[2]*v[2]+u[3]*v[3],
  V_P[j],A_P[j]),j=1..n)];
> plot([seq(zip((u,v)->[u,v],Tau,Vykon[j]),j=1..n)],
  labels=["t","P"],thickness=3,color=red);

```




Obr. 14 Průběh velikosti potřebného příkonu k pohybu shrnovacího ramene na konci jednotlivých prstů v průběhu jedné otáčky

Vezmou se v úvahu pouze kladné hodnoty příkonu, záporné hodnoty se spotřebovávají vlivem tření. Finální graf znázorní celkový potřebný příkon na všech prstech dohromady.

```

> VYKON:=[seq(sum(Vykon[i][j],i=1..n),j=1..Np+1)];
> plot(zip((u,v)->[u,v],Tau,map(u->`if`(u<0,0,u),VYKON)),
  thickness=3,title="Celkový vykon na tyči",labels=["t","P"]);

```



Obr. 15 Pouze kladný příkon potřebný k pohonu shrnovacího ramene



Na závěr se vypočtou délky pracovních drah jednotlivých prstů, které jsou nutné pro stanovení práce třecí síly, která brání přesunu řádkovaného materiálu. Z důvodů přehlednosti je uveden zmenšený analytický výstup programu MAPLE pro 1. prst a numerické hodnoty pro všechny pracovní prsty.

```
> for j from 1 to n do;
  s[j]:=Int(((sqrt(V_prst[j][1]^2+V_prst[j][2]^2+
  V_prst[j][3]^2))),t=t1..t2);
  S[j]:=evalf(subs(DSu,KSu,floor(t)=0,t1=0,t2=1/4,s[j]))+
  evalf(subs(DSu,KSu,floor(t)=0,t1=3/4,t2=1,s[j]));
end do;

> 's[1]'=s[1];
'S[1]'=S[1]; 'S[2]'=S[2]; 'S[3]'=S[3]; 'S[4]'=S[4];
```

$$s_1 = \int_{t_1}^{t_2} \left( \frac{2l \sin\left(\frac{2\pi t}{T}\right) \pi \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right)}{T} + \frac{l \cos\left(\frac{2\pi t}{T}\right) \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)}{r \sqrt{1 - \frac{z(t)^2}{r^2}}} + \frac{2L \cos\left(\frac{2\pi t}{T}\right) \pi}{T} \right.$$

$$\left. - \frac{2(R+r) \sin\left(\frac{2\pi t}{T}\right) \pi}{T} \right)^2 + \left( v - \frac{2l \cos\left(\frac{2\pi t}{T}\right) \pi \sin\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right)}{T} + \frac{l \sin\left(\frac{2\pi t}{T}\right) \cos\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)}{r \sqrt{1 - \frac{z(t)^2}{r^2}}} + \frac{2L \sin\left(\frac{2\pi t}{T}\right) \pi}{T} \right.$$

$$\left. + \frac{2(R+r) \cos\left(\frac{2\pi t}{T}\right) \pi}{T} \right)^2 + \frac{l^2 \sin^2\left(\arccos\left(\frac{z(t)}{r}\right) + \frac{\pi}{4}\right) \left(\frac{d}{dt} z(t)\right)^2}{r^2 \left(1 - \frac{z(t)^2}{r^2}\right)} dt^{(12)}$$

$$S_1 = 5.606154700 \quad S_2 = 5.347896084 \quad S_3 = 5.096462525 \quad S_4 = 4.853020686$$

## VÝSLEDKY A DISKUZE

Popisovaný výpočet matematického modelu shrnovače píce Fella TS 1602 není rozsáhlý, je poměrně jednoduchý, ale přesto umožňuje zjistit základní dynamické parametry.

Jak je možné zjistit z obrázku 12 - Absolutní velikost rychlosti konců prstů v průběhu jedné otáčky – tak rychlost shrnovacích prstů se dost výrazně mění. Nejsou zde sice vidět ostré hroty skokových změn, ovšem zde znázorněný průběh velikosti rychlosti se jim velmi blíží. Z tohoto průběhu je možné odvodit, že nosník shrnovacích prstů - shrnovací tyč bude namáhána na krut a díky tomu, že celý shrnovací rotor se při práci otáčí a nahrnuje materiál, bude toto namáhání kombinováno s ohybem.

Z obrázku 13 - Absolutní velikost zrychlení konců prstů v průběhu jedné otáčky je možné vypořadovat skokové změny velikosti zrychlení konců prstů. Takto velké skokové

změny velikosti zrychlení –  $<200 \text{ ms}^{-2} \sim 20g$  kladou extrémní nároky na uchycení odnímatelných shrnovacích ramen k vlastnímu rotoru.

Z těchto výsledků je možné odvodit, že nejnamáhanějšími částmi bude spojení odnímatelných shrnovacích ramen s vlastním rotorem, nebo uchycení vedení řídicí vačky. Odpovídají tomu i zkušenosti získané provozem shrnovače píce Fella TS 1602 v zemědělských podnicích. Z nejčastějších poruch vyplývá, že konstrukce stroje je v klíčových místech, jako je uchycení snímatelných ramen k vlastnímu rotoru nebo uložení vedení řídicí vačky, velmi poddimenzována. Při provozu dochází k praskání hliníkového odlitku vedení řídicí vačky, zlomení uchycení snímatelných ramen nebo k protahování a vytloukání zajišťovacího otvoru a protilehlého třmenu.

Předložený výpočet nám umožňuje zjistit a upozornit na slabá konstrukční místa tohoto stroje, zároveň jej lze použít pro konstrukční výpočty potřebných úprav.

## **ZÁVĚR**

Výše uvedený postup umožňuje provést všechny výpočty potřebné ke stanovení jak kinematických, tak dynamických charakteristik základních pracovních částí stroje. Z důvodů úspory místa a zvýšení přehlednosti nebyly uvedeny všechny výstupy programu. Nicméně je zcela zřejmé, že použitím uvedených postupů je možné výrazně zefektivnit vývojové a konstrukční práce spojené s projektováním nových typů zemědělských mechanismů.

## LITERATURA

ŠMÍD, V. *Žací stroje s úpravou pokosu*. Brno : MZLU Brno, 2007. Vedoucí diplomové práce doc. Ing. Jan Červinka, CSc.

HAKL, Z. Matematické modelování biologických a technologických procesů v zemědělství. Disertační práce. MZLU: MZLU Brno, 2005. 95 s.

BARTOŇ, S. *The Kinematics of Agricultural Machines*.

<http://www.mapleapps.com/categories/engineering/mechanical/html/agriculture.html>

JAVOREK, F. Technika pro senážování, In. *Zemědělec*, březen 2008, č. 11, s. 24 – 30

ČERVINKA, J. *Stroje pro sklizeň píce na seno*. 2. upravené vyd. Praha : ÚZPI, 2002. ISBN 80-7105-054-7

MAPLESOFT. *Maple10 – User Manual*. Waterloo Maple, 2005, ISBN 1-894511-75-1

Propagační materiály firmy FELLA